The Ballerina programming language
Draft: 2021-01-29

James Clark

Introduction

- Ballerina is a large and complex programming language
- Designed to have an easy-to-learn subset
- Designed to leverage familiarity with existing C-family languages
- But Ballerina Language Specification is not suitable for learning
- Ballerina language is part of a platform, including extensive libraries: this presentation is on language only, not the platform
- This presentation is focused on describing what Ballerina is, not explaining the rationale for the design
- Describes 2021 version of Ballerina ("Swan Lake")

Presentation has three parts

1. How Ballerina provides basic functionality common to most programming languages
2. The functionality that makes Ballerina worthwhile
3. More depth on the functionality described in Part 1

Part 1
How Ballerina does what all programming languages do

Basic functionality

- There's a basic level of general-purpose functionality that is common to almost all programming languages
- Each programming language provides it in a slightly different way: this part describes Ballerina's way of doing this
- This part describes a subset of Ballerina chosen to leverage familiarity with existing languages

Programs and modules

```
import ballerina/io;

public function main() {
   io:println("Hello world!");
}
```

- Program consists of modules
- Modules are one or more `.bal` files
- Modules define named functions
- Module names look like *`org/x.y.z`*
- Standard library uses `ballerina` org
- `import` binds prefix to module name
- Prefix defaults to last part of module name
- Override with `import` *`org/x`* `as` *`m`*
- *m*:*f* means function *f* in module bound to prefix *m*
- `main` function is the program entry point
- `public` makes function visible outside module

Variables and types

```
import ballerina/io;

string greeting = "Hello";

public function greet() {
   string name = "James";
   io:println(greeting, " ", name);
}
```

- Modules and functions can declare variables
- A variables has a type, which constrains what values the variable can hold
- There is a built-in set of named types, including `int`, `float`, `boolean`, `string`
- Assignments are statements not expressions

Functions

```
function add(int x, int y) returns int {
   int sum = x + y;
   return sum;
}
```

- Parameters are declared as in C
- Not allowed to assign to parameters
- `return` statement returns value
- `returns` keyword specifies type of return value
- Function body contain statements

Syntax

```
// This is a comment
// You can have Unicode identifiers
function พิมพ์ชื่อ(string ชื่อ) {
   io:println(ชื่\u{E2D});
}

string 'string = "xyz";
```

- C-like syntax
- Module definitions/declarations and statements either use braces are terminated by semicolon
- Semicolons are not optional
- Comments are `//` to end of line
- Identifier syntax like C
  - Keywords are reserved
  - Prefix reserved keyword with single quote
  - Prefix non-identifier character with \
  - Use \u{H} to specify character using Unicode code point in hex
  - Unicode characters also allowed

Integers

```
int m = 1;

int n = 0xFFFF;
```

- `int` type is 64-bit signed integers (same as `long` in Java)
- Integer literals can be hexadecimal (but not octal)
- Usual arithmetic operators: `+ - * / %`
- Operator precedence as in C
- Do *not* have increment or decrement operators
- Have compound assignment operations e.g. `+=`, `-=`
- Integer overflow is a runtime error
- Usual comparison operators: `== != < > <= >=`

Floating point numbers

```
float x = 1.0;

float y = x + <float>n;
```

- `float` type is IEEE 64-bit binary floating point (same as `double` in Java)
- Same operators as int
- No implicit conversion between integer and floating point values
- Use `<`*T*`>` for explicit conversions
- NaN is `==` to itself: `==` and `!=` on `float` test for same value not IEEE numerically equal

Booleans and conditionals

```
boolean flag = true;

// conditional expression
int n = flag ? 1 : 2;

function foo() {
   if flag {
      io:println(1);
   } else {
      io:println(2);
   }
}
```

- `boolean` type has two values: `true`, `false`
- Conditional expressions use C syntax
- Curly braces are required in if/else and other compound statements
- Parentheses are optional before curly braces
- `!` operator works on booleans only
- && and || operators short-circuit as in C

Nil

```
// value of v can be an int or ()
int? v = ();

// value of s cannot be ()
int n = v == () ? 0 : v;

// ?: operator
int n = v ?: 0;

function foo() {
}

function foo() returns () {
  return ();
}
```

- Ballerina's version of null is called nil and written `()`
- Types do not implicitly allow nil
- Type *T?* means *T* or nil
- Use `==` and `!=` to test whether a value is nil: no implicit conversion to boolean
- Elvis operator x ?: y returns x if it is not nil and y otherwise
- Leaving off return type is equivalent to `returns ()`
- Falling off the end of a function or `return` by itself is equivalent to `return ()`

Strings

```
string grin = "\u{1F600}";

string greeting = "Hello" + grin;
```

- `string` type is immutable sequence of zero or more Unicode characters
- == if sequence has same characters
- String literals use double quotes
  - Usual C escapes e.g. `\n` `\t`
  - Numeric escapes specify Unicode code point using one or more hex digits `\u{`*H*`}`
- Concatenation uses + operator
- No separate character type: a character is represented by string of length 1
- `s[i]` accesses character at index `i` (zero-based)
- `<` `<=` `>` `>=` work by comparing code points
- Unpaired surrogates are not allowed

Langlib functions

```
string s = "abc".substring(1, 2);

// n will be 1
int n = s.length();

// Same as
int n = string:length(s);
```

- Langlib is small library defined by language providing fundamental operations on built-in datatypes
- Langlib functions can be called using convenient method-call syntax
- But these types are *not* objects!
- ballerina/lang.T module for each built-in type T
- Automatically imported using T prefix
- Standard library extends this with rich collection of modules: not part of this presentation

Arrays

```
int[] v = [1, 2, 3];
int n = v[0];
// result will be 3
int len = v.length();
```

- *T*`[]` is an array of *T*
- v[i] does indexed access
- Arrays are mutable: v[i] is an lvalue
- == and != on arrays is deep: two arrays are equal if they have the same members in the same order
- Ordering is lexicographical based on ordering of members
- Langlib arr.length() function gets the length; arr.setLength(n) sets the length

`foreach` statement

```
function sum(float[] v) returns float {
   float r = 0.0;
   foreach float x in v {
      r += x;
   }
   return r;
}
```

```
function sum(float[] v) returns float {
   float r = 0.0;
   foreach int i in 0 ..< v.length() {
      r += v[i];
   }
   return r;
}
```

- `foreach` iterates over an array, by binding a variable to each member of the array in order
- `m ..< n` creates a value that when iterated over will give the integers starting from `m` that are < n
- `foreach` also works for strings, and will iterate over each character of the string

`while` statement

```
type LinkedList record {
   string value;
   LinkedList? next;
};

function len(LinkedList? ll)
                  returns int {
   int n = 0;
   while ll != () {
      n += 1;
      ll = ll.next;
   }
   return n;
}
```

- More flexible iteration than `foreach`
- Usual `break` and `continue` statements

Binary data

```
byte[] data = base64`
   yPHaytRgJPg+QjjylUHakE
   wz1fWPx/wXCW41JSmqYW8=
`;

int x = 0xDEADBEEF;

// OK because byte & int
// will be byte
byte b = x & 0xFF;
```

- Binary data is represented by arrays of byte values
- Special syntax for byte arrays in base 64 and base 16
- Relationship between byte and int not the same as what you are used to
- A `byte` is an `int` in the range 0 to 0xFF
- `byte` is a subtype of `int`
- `int` type supports normal bitwise operators: `& | ^ ~ << >> >>>`
- Ballerina knows the obvious rules about when bitwise operations produce a byte

Maps

```
map<int> m = {
   "x": 1,
   "y": 2
};

int? v = m["x"];

m["z"] = 5;

// m["x"] wouldn't work because
// type would be int? not int
m["z"] = m.get("x");
```

- map<T> is a map from strings to T
- map syntax like JSON
- m[k] gets entry for k; nil if missing
- Use m.get(k) when you know that there's an entry for k
- Maps are mutable: m[k] is an lvalue
- `foreach` will iterate over values of the map
- Iterate over keys by using m.keys() to get the keys as an array of strings

Type definitions

```
type MapArray map<string>[];
MapArray arr = [
   {"x": "foo"},
   {"y": "bar"}
];
```

- Type definition gives a name for a type
- Name is just an alias for the type, like typedef in C

Records

```
record { int x; int y; } r = {
  x: 1,
  y: 2
};

type Coord record {
  int x;
  int y;
};

Coord c = { x: 1, y: 2 };

int x = c.x;
```

- A record type has specific named fields
- Access fields with r.x
- Records are mutable: r.x is an lvalue
- Construct using similar syntax to a map
- Typically combined with type definition
- As usual, name of type is not significant: record is just a collection of fields

Structural typing

- Typing in Ballerina is structural: a type describes a set of values
- Semantic subtyping: subtype means subset
- Universe of values is partitioned into "basic" types
    - each value belongs to exactly one basic type
    - can think of each value as being tagged with its basic type
- There is complexity in making structural typing work with mutation

Immutable basic types (so far):

- nil
- boolean
- int
- float
- string

Mutable basic types (so far):

- array
- map and record

Unions

```
type StructuredName record {
   string firstName;
   string lastName;
};
type Name StructuredName|string;

function nameToString(Name nm)
                  returns string {
   if nm is string {
      return nm;
   }
   else {
      return nm.firstName
          + " " + nm.lastName;
   }
}
```

- $T_1|T_2$ is the union of the sets described by T_1 and T_2
- T? is completely equivalent to T|()
- Unions are untagged
- `is` operator tests whether value belongs to type
- `is` operator in condition causes declared type to be narrowed

Error reporting

```
function parse(string s)
              returns int|error {
   int n = 0;
   int[] cps = s.toCodePointInts();
   foreach int cp in cps {
      cp -= 0x30;
      if cp < 0 || cp > 9 {
         return error("not a digit");
      }
      n = n*10 + cp;
   }
   return n;
}
```

- Ballerina does not have exceptions
- Errors are reported by functions returning error values
- `error` is its own basic type
- An error value includes a string message
- Return type will be union with error
- Return type of `error?` used when the only values explicitly returned are errors
- Error value includes stack trace from point where error(msg) is called
- Error values are immutable
- Ballerina requires ignoring an error to be explicit

Error handling

```
// Convert bytes to a string
// and then to an int
function intFromBytes(byte[] bytes)
            returns int|error {
   string|error ret
     = string:fromBytes(bytes);
   if ret is error {
      return ret;
   }
   else {
      return int:fromString(ret);
   }
}
```

- Usually a function handles errors by passing them up to its caller
- `main` can return an error
- Can use `is` operator to distinguish errors from others value
- There's a shorthand for this pattern

check expression

```
// Convert bytes to a string
// and then to an int
function intFromBytes(byte[] bytes)
           returns int|error {
   string str =
      check string:fromBytes(bytes);
   return int:fromString(str);
}
```

- `check` *E* is used with expression *E* that might result in an error
- If *E* does result in an error, then `check` makes the function return that error immediately
- Type of `check` *E* does not include error
- Control flow remains explicit

Error subtyping

```
type XErr distinct error;
type YErr distinct error;
type Err XErr|YErr;

Err err = error XErr("Whoops!");

function desc(Err err)
                  returns string {
   return err is XErr ? "X" : "Y";
}
```

- `distinct` creates new subtype
- Use subtype when constructing error value
- Works like a nominal type: `is` operator to can distinguish distinct subtypes
- Each occurrence of `distinct` has a unique identifier, used to tag instances of the type

Panics

```
// n must not be 0
function divide(int m, int n)
                     returns int {
   if n == 0 {
      panic error("division by 0");
   }
   return m/n;
}
```

- Ballerina distinguished normal errors from abnormal errors
- Normal errors are handled by returning error values
- Abnormal errors are handled using `panic` statement
- Abnormal errors should typically result in immediate program termination
  - programming bug
  - out of memory
- A panic has an associated error value

any type

```
any x = 1;

// can cast any to specific type
int n = <int>x;

// can convert to string
string s = x.toString();

// can test its type with
// is operator
float f = x is int|float
          ? <float>x
          : 0.0;
```

- `any` means any value except an error
- Equivalent to a union of all non-error basic types
- Use `any|error` for absolutely any value
- Langlib `lang.value` module contains functions that apply to multiple basic types

Ignoring return values

```
// allowed only if return value is ()
doX();

// allowed if return value does not
// include error
_ = getX();

// use checkpanic if you don't want
// to handle an error
checkpanic tryX();
```

- Ballerina does not allow silently ignoring return values
- To ignore a return value, assign it to _
- _ is like an implicitly declared variable of type `any` that cannot be referenced: cannot use _ to ignore an error
- checkpanic is like check, but panics on error rather than returning

Covariance

```
int[] iv = [1, 2, 3];

any[] av = iv; // OK

function foo() {
   // runtime error; otherwise
   // iv[0] would have wrong type
   av[0] = "str";
}
```

- Arrays and maps are covariant
- Allowed to e.g. assign int[] to any[]
  - set of values allowed by int is subset of set of values allowed by any
  - set of values allowed by int[] is subset of set of values allowed by any[]
- Static type-checking guarantees that result of a read from a mutable structure will be consistent with static type
- Covariance means that a write to a mutable structure result in a runtime error
- Arrays, maps and records have "inherent" type that constrains mutation

Object

```
function demoMyClass() {
   m:MyClass x = new m:MyClass(1234);
   x.foo();
   int n = x.n;
};
```

- Separate basic type
- An `object` value has named methods and fields
- Methods and fields are in the same symbol space
- A class both defines an object type and provides a way to construct an object
- Apply `new` operator to a class to get an object
- Call method using obj.foo(args)
- Access field using obj.x

Defining classes

```
public class Counter {
   private int n;

   public function init(int n = 0) {
      self.n = n;
   }

   public function get() returns int {
      return self.n;
   }

   public function inc() {
      self.n += 1;
   }
}
```

- Module can contain class definitions
- `init` method initializes the object
- Arguments to new are passed as arguments to init
- methods use `self` to access their object
- `private` means accessible only by code within the class definition

`init` return type

```
class File {
   string path;
   string contents;
   function init(string p)
                     returns error? {
      self.path = p;
      self.contents =
         check io:fileReadString(p);
   }
};

File f = check new File("test.txt");
```

- init function has a return type, which must be subtype of error?
- If init returns (), then new returns the newly constructed object
- If init returns an error, then new returns that error
- If init does not specify a return type, then return type defaults to () as usual, meaning that `new` will never return error

Identity

```
MyClass obj1 = new MyClass;
MyClass obj2 = new MyClass;

// true
boolean b1 = (obj1 === obj1);
// false
boolean b2 = (obj1 === obj2);
// true
boolean b3 = ([1,2,3] == [1,2,3]);
// false
boolean b4 = ([1,2,3] === [1,2,3]);
// true
boolean b5 = (-0.0 == +0.0);
// false
boolean b6 = (-0.0 === +0.0);
```

- `===` and `!==` operators test for identity
- Identical for mutable basic types means stored at the same address
- `==` and `!=` are not defined for objects
- -0.0 and +0.0 are equal but not identical

const and final

```
const MAX_VALUE = 1000;
const URL = "https://ballerina.io";

final string msg = loadMessage();
```

- const means read-only and known at compile-time
- Type is singleton: set containing single value
- Variable or class field can be declared as final, meaning cannot be assigned to after it has been initialized

Enumerations

```
enum Color {
   RED, GREEN, BLUE
}

// shorthand for
const RED = "RED";
const GREEN = "GREEN";
const BLUE = "BLUE":
type Color RED|GREEN|BLUE;

enum Color {
   RED = "red",
   GREEN = "green",
   BLUE = "blue"
};
```

- Enumerations are shorthand for unions of string constants
- A const can be used as a singleton type
- Not a distinct type
- Can specify string constants explicitly

`match` statement

```
const KEY = "xyzzy";

function mtest(any v) returns string {
   match v {
      17 => { return "number"; }
      true => { return "boolean"; }
      "str" => { return "string"; }
      KEY => { return "constant"; }
      0|1 => { return "or"; }
      _ => { return "any"; }
   }
}
```

- Like `switch` statement in C, JavaScript
- Matches value not type
- == is used to test whether left hand side matches value being matched
- Left hand side can be
  - simple literal (nil, boolean, int, float, string)
  - identifier referring to a constant
- Left hand side of _ matches if value is of type any
- Use | to match more than one value

Type inference

```
var x = "str";

function printLines(string[] sv) {
  foreach var s in sv {
     io:println(s);
  }
}

// Infer x as type MyClass
var x = new MyClass;

// Infer class for new as MyClass
MyClass x = new;
```

- Type inference is local: restricted to single expression
- Goal is: Do Not Repeat Yourself
- `var` says that type of variable from type of expression used to initialize it
  - Convenient with foreach statement
- Also infer type of value to be created from type of variable
- Overuse can make code harder to understand

Functional programming

```
var isOdd = function(int n) returns boolean {
   return n % 2 != 0;
}

type IntFilter function(int n) returns boolean;

function isEven(int n) returns boolean {
  return n % 2 == 0;
}

IntFilter f = isEven;

int[] nums = [1, 2, 3];

int[] evenNums = nums.filter(f);

int[] oddNums = nums.filter(n => n % 2 != 0);
```

- Functions are values
- Separate basic type
- Anonymous function and type syntax look like function definition without the name
- Arrays provide the usual functional methods: filter, map, forEach, reduce
- Like `foreach`, also work on maps and strings
- Shorthand syntax for when type is inferred and body is an expression

Asynchronous function calls

```
// assume foo() returns int
future<int> f1 = start foo();
future<int> f2 = start foo();


int x1 = check wait f1;
int x2 = check wait f2;

// or better...
var f = wait { f1, f2 };
int x1 = check f.x1
int x2 = check f.x2
```

- `start` calls a function asynchronously
- Runs on separate logical thread ("strand"): cooperatively multitasked by default
- Result will be of type future<T>
- `future` is a separate basic type
- wait for future<T> gives T|error (waiting for the same future more than once gives an error)
- Can wait for multiple futures
- Use f.cancel() to terminate a future

Documentation

```
# Adds two integers.
# + x - an integer
# + y - another integer
# + return - the sum of `x` and `y`
public function add(int x, int y)
                     returns int {
  return x + y;
}
```

- Lines starting with # contain structured documentation
- Uses Markdown format with some additional conventions

Annotations

```
// The @display annotation applies
// to the transform function
@display {
   iconPath: "transform.png"
}
public function transform(string s)
   returns string {
   //...
}

// annotation on start
future<int> f = @strand {
                    thread: "any"
                }
                start foo();
```

- Annotations start with `@tag`
- Annotations come before what they apply to
- Unprefixed tags refer to standard platform-defined annotations
- Prefixed tags refer to annotations declared in modules
- `@tag` can be followed by record constructor expression

Part 2
What makes Ballerina worthwhile

Network interaction

- Consuming services
- Providing services

Consuming services: client objects

`Example`

- Client objects provide `remote` methods, which are used to interact with a remote service
- A client object is created by applying new to a client class
  - Defined by `client class {...}`
- Applications typically do not need to write client classes, which are either
  - provided by library modules
  - generated from some flavour of IDL
- Remote method calls use `->` syntax
  - support sequence diagram view
  - not allowed nested within expressions
  - separate symbol space for method names
  - remote methods implicitly public

Providing services

- Service object
  - remote methods defined by application; no need to define a class
  - attached to a Listener object
- Listener
  - receives network input
  - makes calls to remote methods of attached service objects
  - registered with module
- Module
  - initialized on program startup
  - starts up registered Listeners after initialization
  - shuts down registered Listener during program shutdown

Listener declaration

```
import ballerina/http;

listener h = new http:Listener(8080);
```

- Allowed at module level
- Like a variable declaration, but registers the newly created Listener object with the module
- If `new` returns an error, then module initialization fails

Module lifecycle

- All modules are initialized at program startup
- A module's listeners are registered during module initialization
- Module initialization is ordered so that if module A imports module B, then module A is initialized after module B
- Initialization phase ends by calling `main` function
- If there are registered listeners, then initialization phase is followed by listening phase
- Listening phase starts by calling `start` method on each registered listener
- Listening phase is terminated by signal (e.g. SIGINT, SIGTERM)
- Calls either `gracefulStop` or `immediateStop` on each registered listener

Module `init` function

```
import myService as _;

function init() {
   io:println("Hello world");
}
```

- A module can have an `init` function just like an object
- Initialization of a module ends by calling its `init` function if there is one
- Return type must be a subtype of `error?`
- Usually it's an error to import a module without using it
- If you want to import a module because of what its initialization does (e.g. registering services), then use `as _` in the import

Constructing objects without classes

```
var obj = object {
   function greet() returns string {
      return "Hello world";
   }
};

string greeting = obj.greet();
```

- An object can be constructed directly, without defining a class

Service declaration

```
import ballerina/udp;

service on new udp:Listener(8080) {
   remote function onDatagram(Datagram dg) {
      io.println("bytes received: ",
                 dg.bytes.length());
   }
};
```

- Evaluates an expression to create a listener object
- Registers listener, if necessary
- Creates service object using object constructor
- Attaches service object to the listener
- Type of Listener determines required type of remote methods
- Annotations are used extensively e.g. for security

Service declaration desugaring

```
service on new udp:Listener(8080) {
   remote function onDatagram(Datagram dg) { ... }
};

// desugars to

listener u = new udp:Listener(8080);

var obj = service object {
   remote function onDatagram(Datagram dg) { ... }
};

function init() {
   u.attach(obj);
}
```

Resource concept

- Service objects use remote methods to expose services in procedural style: remote methods are named by verbs
- Service objects use resources to expose services in an RESTful style: resources are named by nouns
- Resources are motivated by HTTP, but are generic enough also to work for GraphQL also

Resources

- Resource method associated with combination of accessor and resource name
- Accessors determined by network protocol
- Network-oriented generalization of OO getter/setter concept
- Service declaration specifies base path for resource names
- In HTTP, function parameters come from query parameters

```
service / on new http:Listener(8080) {
   resource function get hello(string name) returns string {
      return "Hello, " + name;
   }
}
```

Hierarchical resources

- Resource name is relative path, which can have multiple path segments
- Base path is absolute path
- Single listener can have multiple services each with different base paths

```
service /demo on new http:Listener(8080) {
   resource function get greeting/hello(string name) returns string {
      return "Hello, " + name;
   }
}
```

Resource path parameters

- Path segments can be parameters

```
// GET /demo/greeting/James would return "Hello, James"

service /demo on new http:Listener(8080) {
   resource function get greeting/[string name]() returns string {
      return "Hello, " + name;
   }
}
```

Hierarchical services

- Resource methods can return service objects
- Semantics is that path of resource method becomes base path of service object: similar to filesystem mount
- Root service is special case of this
- Basis for GraphQL support: each GraphQL object is represented by a service object

Plain data

Plain data

- Ballerina has concept of "plain data": data that is independent of any specific code operating on the data
- Network interfaces between programs are based on plain data
- Opposite of objects, which combine data and code
- Plain data supports deep copy and deep equality
- Plain data supports serialization/deserialization without coupling
- Key goal of Ballerina is to facilitate programs that work on plain data

Ballerina basic types

Simple types

Always plain data

- nil
- boolean
- int
- float
- *decimal*

Sequence

Always plain data

- string
- *xml*

Structural

Plain data if members are

- array/*tuple*
- map/record
- *table*

Behavioural

Not plain data

- function
- object
- error
- *stream*
- *typedesc*
- *handle*

`decimal` type

```
function floatSurprise() {
  float f = 100.10 - 0.01;
  // will print 100.08999999999999
  io:println(f);
}

decimal nanos = 1d/1000000000d;
```

- Third numeric type
  - works like int and float
  - no implicit conversion
- Represents decimal fractions exactly
- Avoids surprises that you get with float
- Preserves precision: 2.1kg and 2.10kg don't mean the same to humans
- Separate basic type; counts as anydata
- Literal uses `d` suffix (`f` suffix is for float)
- Floating point, not infinite precision
  - 34 decimal digits
  - 22 digits enough for US national debt in ¢
  - 27 digits enough for age of universe in ns
- No infinity, NaN or negative zero

Plain data basic types to come

- table
  - works uniformly with array and map
  - table contains records
  - support access by key using concept similar to primary keys in SQL
  - fields containing key are read-only
- xml
  - sequence of xml items (element, text, processing instruction, comment)
  - sequence concept similar to string and to XQuery
  - XML attributes represented as `map<string>`
  - xml literals support XML syntax

Immutability

- anydata values can be made immutable
- Simple and string values are inherently immutable
- A structural value can be constructed as mutable or immutable
  - It has a read-only flag as part of its value
  - Read-only flag is fixed at the time of construction
  - Attempting to mutate a read-only structure causes a panic at runtime
- Immutability is deep: an immutable structure can only have immutable members
  - an immutable value is safe for concurrent access without locking

`anydata` type

```
anydata x1 = [1, "string", true];
anydata x2 = x1.clone();

// true
boolean eq = (x1 == x2);

const RED = {R: 0xFF, G: 0, B: 0};
```

- Type for plain data is `anydata`
- Subtype of `any`
- == and != operators test for deep equality
- `x.clone()` returns deep copy, with same mutability
- `x.cloneReadOnly()` returns deep copy that is immutable
- Both `x.clone/cloneReadOnly()` do not copy immutable parts of x
- `const` structures are allowed
- Equality and cloning handle cycles

User-defined types describe both data in memory and data on the wire

Optional fields

```
type Headers record {
   string from;
   string to;
   string subject?;
};

Header h = {
  from: "John",
  to: "Jill"
};

string? subject = h?.subject;
```

- Records can have optional fields
- Use `?.` operator to access optional field

Open records

```
type Person record {
   string name;
};
type Employee record {
   string name;
   int id;
};
Employee e = {
   name: "James", id: 10
};
Person p = e;
Person p2 = {
   name: "John", "country": "UK"
};
map<anydata> m = p2;
```

- Record types are by default open: they allow fields other than those specified
- Type of unspecified fields is `anydata`
- Records are maps
- Open records belongs to `map<anydata>`
- Use quoted keys for fields not mentioned in the record type

Controlling openness

```
type Coord record {|
   float x;
   float y;
|};
Coord x = { x: 1.0, y: 2.0 };
map<float> m = x;

type Headers record {|
   string from;
   string to;
   string...;
|};
Headers h = {
   from: "Jane", to: "John"
};
map<string> m = h; // OK
```

- Use `record {| ... |}` to describe a record type that allows exclusively what is specified in the body
- Use *`T...`* to allow other fields of type T
- `map<`*`T`*`>` same as `record {| `*`T`*`...; |}`

`json` type

```
import ballerina/lang.value;

json j = { "x": 1, "y": 2 };

string s = j.toJsonString();

json j2 =
  check value:fromJsonString(s);
```

- `json` type is a union: `()|boolean|int|float|decimal |string |json[]|map<json>`
- A `json` value can be converted to and from JSON format straightforwardly
  - except for choice of Ballerina numeric type
- Ballerina syntax is compatible with JSON
- `json` is `anydata` without `table` and `xml`
- `toJson` recursively converts `anydata` to `json`
  - table values are converted to arrays
  - xml values are converted to strings
- XML and JSON are not uniform

Working with JSON: two approaches

- Approach 1: Work with json values directly
- Approach 2: Work with application-specific, user-defined subtype of `anydata`
  - Convert from JSON to application-specific type
  - Process using application-specific subtype
  - Convert back to JSON from application-specific type
- Ballerina supports both approaches
- Ballerina's strength is making Approach 2 really easy

Working with json directly

```
json j = {
   x: {
      y: {
         z: "value"
      }
   }
}

json v = check j.x.y.z;
string s = check v;
// short for
string s =
   check value:ensureType(v, string);

// put it together
string s = check j.x.y.z;
```

- json values use "lax" typing
- Expressions that would usually be a compile-time error instead result in an error at runtime
- User experience similar to dynamic language
- Two cases
  - accessing a field with j.x or j?.x
  - implicit conversion from json value to unstructured type
- ensureType performs numeric conversions

match statement with maps

```
function foo(json j) returns error? {
   match j {
      { command: "add", amount: var j }
      => {
         decimal n = check j;
         add(n);
      }
      _ => {
         return error("invalid command");
      }
   }
```

- match statement can be used to match maps
- Useful for working directly with json
- Match semantics are open (may have fields other than those specified in the pattern)

Converting from user-defined type to JSON

```
// closed type
type Coord record {|
  float x;
  float y;
|};
Coord coord = { x: 1.0, y: 2.0 };
// nothing to do
json j = coord;
// If coord is is open:
type Coord record {
  float x;
  float y;
};
// usually happens automatically
json j = coord.toJson();
```

- Conversion from json value to JSON format is straightforward
- Problem here is converting from application-specific, user-defined subtype of anydata into json
- In many cases, this is a no-op: user-defined type will be subtype of json as well as of anydata
- With tables, xml or records open to anydata, use toJson to convert anydata to json
- APIs that generate JSON typically accept anydata and automatically apply toJson

Converting from JSON to user-defined type

```
// closed type
type Coord record {
  float x;
  float y;
};
json j = { x: 1.0, y: 2.0 };

// Runtime error!
Coord c = <Coord>j;

// This will work
Coord c = <Coord>j.cloneReadOnly();
```

- This way round is more interesting!
- With mutable values, would not be type-safe to allow a cast
- Mutable structures have inherent type that limits mutation
  - does not affect equality
  - clone copies the type
- Cast to T will work on mutable structure s only if inherent type of s is subtype of T
- Casting of read-only value will work, but does not do numeric conversions

Converting to user-defined type: `cloneWithType`

```
type Coord record {
  float x;
  float y;
};
json j = { x: 1.0, y: 2.0 };

Coord c
  = check j.cloneWithType(Coord);

// Argument defaulted from context
Coord c = check j.cloneWithType();
```

- Langlib function in lang.value
- Result recursively uses specified type as inherent type of new value
- Argument is a `typedesc` value
- Static return type depends on argument
- Argument defaulted from context
- Automatically performs numeric conversions as necessary
- Every part of value is cloned, including read-only structures
- Graph structure is not preserved
- Variant `fromJsonWithType` also does reverse of conversions done by toJson

Resource method typing

- Resource method arguments can use user-defined types
- Listener will use introspection to map from protocol format (typically JSON) to user-defined type, using `cloneWithType`
- Return value that is subtype of anydata will be mapped from user-defined type to protocol format, typically JSON, using toJson
- Can generate API description (e.g. OpenAPI) from Ballerina service declaration
- Annotations can be used to refine the mapping between Ballerina-declared type and wire format

Numbers in json

- json type allows int|float|decimal
- toJsonString will convert int|float|decimal into JSON numeric syntax
- fromJsonString converts JSON numeric syntax into int, if possible, and otherwise decimal
- cloneWithType or ensureType will then convert into user's chosen type
- Net result is that you can use json to exchange full range of all three Ballerina numeric types
- -0 is an edge case: represented as float

Query expressions

SQL-like syntax for list comprehensions

```
int[] nums = [1, 2, 3, 4];

// Result is [10, 20, 30, 40]
int[] numsTimes10 =
   from var i in nums
   select i*10;

// Result is [2, 4]
int[] evenNums =
   from var i in nums
   where i % 2 == 0
   select i;
```

- Query-like expressions start with `from` and `end` with select

Destructuring records

```
type Person record {
   string first;
   string last;
   int yearOfBirth;
};
Person[] persons = [];

// Projection with first and last fields
var names =
  from var {first: f, last: l} in persons
  select {first: f, last: l};

// more simply
var names =
  from var {first, last} in persons
  select {first, last};
```

- Particularly useful with query expressions, but works anywhere you can have `var`
- Thing following var is called a binding pattern
- Semantics of binding pattern is open

Let clause

```
string[] names =
   from var {first, last} in persons
   let int len1 = first.length()
   where len1 > 0
   let int len2 = last.length()
   where len2 > 0
   let string name = first + " " + last
   select name;
```

- Query expressions can have `let` clauses
- Can be anywhere between `from` and `select`
- Multiple `where` clauses are allowed
- Semantics similar to XQuery FLWOR

Ordering

```
type Employee record {
   string firstName;
   string lastName;
   decimal salary;
};

Employee[] employees = [...];

Employee[] sorted =
   from var e in employees
   order by e.lastName, e.firstName
   select e;
```

- Ordering works the same as with <, <=, >, >= operators

Limit clause

```
Employee[] top100 =
   from var e in employees
   order by e.salary descending
   limit 100
   select e;
```

- `limit` clause limits number of results from earlier clauses

Tables

Table concept

- Ballerina approach is to have a few general-purpose data structures
  - So far we have seen: array and map
  - There is one more: table
- A table is a collection of records; each record represents a row of the table
- A table is plain data if and only if its rows are plain data
- Records are uniquely identified within the table by their keys
- Keys are stored in record fields, which must be read-only
- Compared to maps:
  - key is part of the value, rather than separate
  - type of key is not restricted to string
  - order is preserved

Table syntax

```
type Employee record {
   readonly string name;
   int salary;
};
table<Employee> key(name) t = table [
  { name: "John", salary: 100 },
  { name: "Jane", salary: 200 }
];

Employee? e = t["Fred"];

function increaseSalary(int n) {
  foreach Employee e in t {
     e.salary += n;
  }
}
```

- A record field can be declared as `readonly`: cannot mutate the field after it record is created
- table type gives type of row and name of key field
- table constructor expression looks like an array constructor
- `foreach` statement will iterate over a table's rows in order
- Use t[k] to access a row using its key

Multiple keys

```
type Employee record {
   readonly string firstName;
   readonly string lastName;
   int salary;
};

table<Employee> key(firstName, lastName) t = table [
  { firstName: "John", lastName: "Smith", salary: 100 },
  { firstName: "Fred", lastName: "Bloggs", salary: 200 }
];

Employee? e = t["Fred", "Bloggs"];
```

Structured keys

```
type Employee record {
   readonly record {
      string first;
      string last;
   } name;
   int salary;
};

table<Employee> key(name) t = table [
   { name: {first: "John", last: "Smith"}, salary: 100 },
   { name: {first: "Fred", last: "Bloggs"}, salary: 200 }
];

Employee? e = t[{first: "Fred", last: "Bloggs"}];
```

- Keys can be structured: any subtype of plain data
- Value of key must be immutable
- Initializer of readonly field will be constructed as immutable
- In other cases, can use `cloneReadOnly` to create immutable structure

Querying tables

```
type Employee record {|
   readonly int id;
   string firstName;
   string lastName;
   decimal salary;
|};

table<Employee> key(id) employees =
  table [...];

int[] salaries =
  from var { salary } in employees
  select salary;
```

- Tables combine nicely with query
- Maps not so much
- By default, queries the output the same kind of structure as their input (what `from` iterates over)

Creating tables with query

```
var highPaidEmployees =
   table key(id)
   from var e in employees
   where e.salary >= 1000
   select e;
```

- Query expressions can create tables
- Key of created table can be specified explicitly

Join clause

```
type User record {|
   readonly int id;
   string name;
|};
type Login record {|
   int userId;
   string time;
|};
table<User> key(id) users = [...];
Login[] logins = [...];

string[] loginLog =
  from var login in logins
  join var user in users
       on login.userId equals user.id
  select user.name + ":" + login.time;
```

- Query can take advantage of table keys by using a join clause
- Does inner equijoin
- Results similar to nested `from` clause and `where` clause
- Implemented as hash join: table keys allow you to avoid building a hash table

Streams

Stream type

- A stream represents a sequence of values that are generated as needed
- When a stream is finished, it returns a termination value, which is error or nil
- Type `stream<T,E>` is a stream where
  - members of the sequence are type T
  - termination value is type E
- `stream<T>` means `stream<T,()>`
- Separate basic type, but like an object

Querying with streams

```
type LS stream<string,io:Error?>;
// strip blank lines
function strip(LS lines) returns LS {
   stream from var line in lines
   where line.trim().length() > 0
   select line;
}

function count(LS lines)
               returns int|io:Error {
   int nLines = 0;
   check from var line in lines
   do {
      nLines += 1;
   }
   return nLines;
}
```

- If stream terminates with error, result of query expression is an error
- from/do; result is error?
- Creating streams: lazily evaluated; check-fail semantics

Backtick templates

```
string name = "James"
// Result is "Hello, James"
string s = string`Hello, ${name}`;
string s = string`Backtick:${"`"}`;
```

- Consists of tag followed by characters surrounded by backticks
  - Can contain expressions in `${...}` to be interpolated
  - No escapes recognized: use expression to escape
  - Can contain newlines
- Processed in two phases
  - Phase 1 does tag-independent parse: result is list of strings and expressions
  - Phase 2 is tag-dependent
- Phase 2 for ``string`...` `` converts expressions to strings and concatenates
- `base16` and `base64` tags do not allow expressions

Raw templates

- A raw template is a backtick template without a tag
- Exposes result of phase 1 without further processing
- Raw template is evaluated by evaluating each expression and creating an object containing
  - an array of the strings separated by insertions
  - an array of the results of expression evaluation and an array of strings separating

```
function getOrders(int customerId) returns stream<Order,sql:Error?>
   return db->query(`SELECT * FROM order
                     WHERE customer_id = ${customerId}`);
}
```

XML

XML overview

- Separate basic type xml
- Uses sequence concept similar to XQuery and XPath2
- Based on XML Infoset, rather than PSVI
- Allows XML syntax to be used to construct xml values
- xml type is designed to work well for HTML as well as XML
- Navigation syntax with XPath-like functionality
- Works with query expressions to provide XQuery FLWOR-like functionality
- No up pointers: elements do not have a reference to parents or siblings

Sequences

- Ballerina has two basic types that are sequences: string, xml
- A value is a sequence of basic type T if it is
  - an empty sequence of basic type T,
  - a singleton of basic type T, or
  - a concatenation of two sequences of basic type T
- Sequences differ from arrays:
  - sequences are flat: no nesting
  - there is no difference between a singleton *x* and a sequence consisting of just *x*
  - basic type of sequence determines basic type of members
- Membership of a sequence is immutable e.g. cannot mutate a sequence of one item into a sequence of two items
- A sequence has no identity: two sequences are === if their members are ===

XML data model

- xml value is a sequence representing the parsed content of an XML element
- xml value has four kinds of item
  - element, processing instruction and comment item correspond 1:1 to XML infoset items
  - text item corresponds to one or more Character Information Items
- XML document is an xml sequence with only one element and no text
- An element item is mutable and consists of:
  - name: type string
  - attributes: type map<string>
  - children: type xml
- A text item is immutable
  - it has no identity: == is the same as ===
  - consecutive text items never occur in an xml value: they are always merged

xml templates

```
string url = "https://ballerina.io";

xml content = xml`
<a href="${url}">Ballerina</a> is
an <em>exciting</em> new language!`;

xml p = xml`<p>${content}</p>`;
```

- xml values can be constructed using an XML template expression
- Phase 2 processing for `xml` template tag parses strings using the XML 1.0 Recommendation's grammar for content (what XML allows between a start-tag and and end-tag)
- Interpolated expressions can be
  - in content, xml or string values
  - in attribute values, string values

xml operations

```
xml x1 = xml`<para id="greeting">Hello</p>`
string id = check x1.id;
```

- + does concatenation
- == does deep equals
- foreach iterates over each item
- x[i] gives i-th item (empty sequence if none)
- x.id accesses required attribute named id: result is error if there is no such attribute or if x is not a singleton
- x?.id accesses optional attribute named id: result is () if there is no such attribute
- Langlib lang.xml provides other operations
- Mutate an element using e.setChildren(x)

xml subtyping

```
xml:Element p = xml`<p>Hello</p>`;

function stringToXml(string s)
                returns xml:Text {
   return xml:createText(s);
}

function rename(xml x, string oldName,
                string newName) {
   foreach xml:Element e in x.elements() {
      if e.getName() == oldName {
         e.setName(newName);
      }
      rename(x.getChildren());
   }
}
```

- An xml value belongs to `xml:Element` if it is consists of just an element item
- Similarly for `xml:Comment` and `xml:ProcessingInstruction`
- An xml value belongs to `xml:Text` if it consists of a text item or is empty
- An xml value belongs to the type `xml<T>` if each of its members belong to T
- Functions in lang.xml use this to provide safe and convenient typing e.g.
  - `x.elements()` returns element items in x as type `xml<xml:Element>`
  - `e.getName()` and `e.setName()` are defined when `e` has type `xml:Element`

XML navigation syntactic sugar

x.<para>	every element in x named para
x/*	for every element e in x, the children of e
x/<para>	for every element e in x, every element named para in the children of e
x/<th|td>	for every element e in x, every element named th or td in the children of e
x/<*>	for every element e in x, every element in the children of e
x/*.text()	for every element e in x, every text item in the children of e
x/**/<para>	for every element e in x, every element named para in the descendants of e
x/<para>[0]	for every element e in x, first element named para in the children of e

Querying with XML

- Can use query expressions to manipulate XML

```
function paraByLang(xml x, string lang) returns xml {
   return from para in x.<para>
          where para?.lang == lang
          select para;
}
```

Combining XML templates and query

- XML templates combine nicely with query e.g. you can have a templates containing a query expression containing a template

```
type Person record {|
   string name;
   string country;
|};
function personsToXml(Person[] persons) returns xml {
   return xml`<data>${
      from var {name, country} in Persons
      select xml`<person country="${country}">${name}</person>`
   }</data>`;
}
```

XML namespaces

```
xml:Element e =
 xml`<p:e xmlns:p="http://example.com/"/>`;

// name will be "{http://example.com}e"
string name = e.getName();
```

- Goal is to support for namespaces, but no added complexity if you don’t use them
- Qualified name `ns:x` in XML is expanded into `{url}x` where `url` is namespace name bound to `ns`
- XML namespace declarations are kept as attributes using standard binding of `xmlns` to `http://www.w3.org/2000/xmlns/`

`xmlns` declarations

```
xmlns "http://example.com" as eg;

xml x = xml`<eg:doc>Hello</eg:doc>`;

xmlns "http://example.com" as ex;

// will be true
boolean b = (x === x.<ex:doc>);

// exdoc will be "{http://example.com}doc"
string exdoc = ex:doc;
```

- `xmlns` declarations are like import declarations, but bind the prefix to a namespace URL rather than a module
- `xmlns` declarations in the Ballerina module provide namespace context for parsing xml templates
- Qualified names in Ballerina modules are expanded into strings using the `xmlns` declarations in the module
- xmlns declarations also allowed at block level

Configurable variables

```
# Port on which to run the service
configurable int port = 8080;

# Password must be supplied in
# configuration file
configurable string password = ?;
```

- A module-level variable can be declared as configurable
- The initializer of a configurable variable can be overridden at runtime (e.g. by a TOML file)
- A variable where configuration is required can use an initializer of `?`

Sequence-diagram based concurrency

Named workers

```
function main() {
   io:println("Initializing");
   worker A {
      io:println("In worker A");
   }
   worker B {
      io:println("In worker B");
   }
   io:println("In default worker");
   wait A;
   io:println("After wait A");
}
```

- Normally all of a function's code belongs to its "default worker", which has a single logical thread of control
- A function can also declare named workers, which run concurrently the default worker and other named workers
- Code before any named workers is executed before named workers starts
- Variables declared before all named workers and function parameters are accessible in named workers

Sequence diagrams

- A function can be viewed as a sequence diagram
- Lifeline (vertical line) for each worker (both named and default)
- Lifeline for each client object parameter or variable in initialization section, representing remote system to which the client object is sending messages
- Each remote method call on a client object is represented as a horizontal line between the lifeline of the worker making the call and the remote system

Waiting for workers

```
function main() {
   io:println("Initializing");
   worker A {
      io:println("In worker A");
   }
   io:println("In default worker");
   wait A;
   io:println("After wait A");
}
```

- Named workers can continue to execute after the default worker terminates and the function returns
- A worker (default or named) can use `wait` to wait for a named worker

Strands

- By default named workers do not run on separate threads
- Each named worker has a "strand" (logical thread of control) and execution switches been strands only at specific points such as
  - doing a wait
  - when a library function invokes a system call that would block
- This avoids the need for users to lock variables that are accessed from multiple named workers
- An annotation can be used to make a strand run on a separate thread

Named worker return values

```
function demo(string s)
                   returns int|error {
   worker A returns int|error {
      int x = check int:fromString(s);
      return x + 1;
   }
   int y = check wait A;
   return y + 1;
}
```

- Named workers have a return type, which defaults to nil
- A return statement in a named worker terminates the worker not the function
- Using check in a named worker will thus
- Waiting on a named worker will give its return value

Named workers and futures

```
function demo() returns future<int> {
   worker A returns int {
      return 42;
   }
   return A;
}

type FuncInt function() returns int;

function startInt(FuncInt f)
                returns future<int> {
   worker F returns int {
      f();
   }
   return F;
}
```

- Futures and workers are the same thing
- A reference to a named worker can be implicitly converted into a future
- `start` is sugar for calling a function with a named worker and returning the named worker as a future

Inter-worker message passing

```
function demo() returns int {
   worker A {
      1 -> B;
      2 -> C;
   }
   worker B {
      int x1 = <- A;
      x1 -> default;
   }
   worker C {
      int x2 = <- B
      x2 -> default;
   }
   int y1 = <- B;
   int y2 = <- C;
   return y1 + y2;
}
```

- Use `-> W` or `<- W` to send a message to or receive a message from worker W (use `default` to refer to default worker)
- Messages are copied using clone(); implies immutable values are passed without copy
- Message sends and receives are paired up at compile-time
- Each pair turns into horizontal line in sequence diagram
- Easy to use and safe, but limited expressiveness

Inter-worker failure propagation

```
function demo() returns int|error {
   worker A returns error? {
      check foo();
      42 -> default;
   }
   int x = check <- A;
   return x;
}
```

- Workers may need to call functions that can return an error
- Pairing up of sends and receives guarantees that each send will be received, and vice-versa, *provided* neither sending nor receiving worker has failed
- Send to or receive from failed worker will propagate the failure

Transactions

Transactions overview

- Ballerina runtime includes transaction manager
- Language support for delimiting transactions
- Current transaction part of execution context
- `transactional` qualifier guarantees there is a current transaction
- Makes it easy to retry when there's an error committing
- Works with client and service objects to support distributed transactions
- Named workers can be transactional
- Memory is not transactional

`check` semantics

```
function demo() returns error? {
   do {
      check foo();
      check bar();
       if !isOK() {
         fail error("not OK");
      }
   }
   on fail var e {
      io:println(e.toString());
      return e;
   }
}
```

- `check` semantics is not simply to return on error
- When check gets an error, it *fails*
  - Enclosing block decide how to handle failure
  - Most blocks pass failure up to enclosing block
  - Function definition handles failure by returning the error
- `on fail` can catch the error
- `fail` statement is like `check` but always fails
- Differs from exceptions in that control flow is explicit

`transaction` statement

```
function transfer(Update[] updates)
                        returns error? {
   transaction {
      foreach var u in updates {
         check exec(u);
      }
      check commit;
   }
}

transactional function exec(Update u)
                        returns error? {
   // run some SQL to do the update
}
```

- `transaction` statement begins a new transaction and executes a block
- If there is a fail or panic in the execution of the block, then the transaction is rolled back
- Commit of a transaction must be done explicitly using `commit`
  - must be lexically within a transaction statement
  - `commit` may return an error; usual rules on not ignoring errors apply
- An exit from a transaction block that does not explicitly commit and is not a panic or fail must explicitly `rollback`

Concurrency safety

`lock` statement

```
int n = 0;

function inc() {
   lock {
      n += 1;
   }
}
```

- Semantics are like an atomic section: execution of outermost lock blocks is not interleaved
- Naive implementation uses single, global, recursive lock
- Efficient implementation can do compile-time lock inference

Service concurrency

- Goal is for Listener to service incoming requests concurrently
- Be able to look at the program and tell whether when it's safe for strands to be executed on separate threads
- Read-only helps, but is not enough: although function values are read-only, this tells you nothing about what executing a function will result in mutating state
- Ballerina provides `isolated` feature to handle this

Isolated functions

```
type R record {
   int v;
};

final int N = getN();

isolated function set(R r) {
   r.v = N;
}

R r = {v: 0};

// This is not isolated
function setGlobal(int n) {
   r.v = n;
}
```

- Informal concept: a call to an isolated function is concurrency-safe **if** it is called with arguments that are safe at least until the call returns
- A function defined as isolated
  - has access to mutable state only through its parameters
  - has unrestricted access to immutable state
  - can only call functions that are isolated
- Constraints are enforced at compile-time
- `isolated` is part of the function type
- Weaker concept than pure function

Combining `isolated` functions and `lock`

- Goal is to allow isolated functions to use lock to access mutable module-level state
- Key concept is isolated root
- A value *r* is an *isolated root* if mutable state reachable from *r* cannot be reached from outside except through *r*
- An expression is an *isolated expression* if it follows rules that guarantee that its value will be an isolated root e.g.
    - a boolean, numeric or string literal is always isolated
    - an expression [E1, E2] is isolated if E1 and E2 are isolated
    - an expression f(E1, E2) is isolated if E1 and E1 are isolated, and the type of f is an isolated function

Isolated variables

```
isolated int[] stack = [];

isolated function push(int n) {
   lock {
      stack.push(n);
   }
}

isolated function pop() returns int {
   lock {
      return stack.pop();
   }
}
```

- When a variable is declared as isolated, compiler guarantees that it is an isolated root and accessed only within a lock statement
- Isolated variable declaration must be module-level, not public, initialized with isolated expression
- A `lock` statement that accesses an isolated variable must maintain isolated root invariant:
  - access only one isolated variable
  - call only isolated functions
  - restricted in how values are transferred into and out of the lock statement
- Isolated functions are allowed to access isolated module-level variables, provided they follow the above rules

Isolated methods

- Object methods can be isolated
- An isolated method is the same as an isolated function with `self` treated as a parameter
- An isolated method call is concurrency-safe if both the object is safe and the arguments are safe
- This is not quite enough for service concurrency: when a Listener makes calls to a remote or resource method,
  - it can ensure the safety of arguments it passes
  - it has no way to ensure the safety of the object itself (since the object may have fields)

Isolated objects

```
isolated class Counter {
   private int n = 0;

   isolated function get()
                  returns int {
      lock {
         returns self.n;
      }
   }

   isolated function inc() {
      lock {
         self.n += 1;
      }
   }
}
```

- An object defined as isolated is similar to a module with isolated module-level variables
- Mutable fields of an isolated object
  - must be private and so can only be accessed using `self`
  - must be initialized with an isolated expression
  - must only be accessed within a `lock` statement
  - `lock` statement must follow the same rules for `self` as for an isolated variable
  - mutable means either not final or has type that is not readonly
- Isolated root concept treats isolated objects as opaque
- Isolated functions can access a final variable whose type is an isolated object

Inferring `isolated`

- `isolated` is a complex feature, which would be a lot for an application developer to understand
- A typical Ballerina application consists of a single module that imports multiple library modules
- Within a single module, we can infer `isolated` qualifiers
- Object w/o mutable fields is inherently isolated
- Application developer's responsibility is to use `lock` statement where needed
  - access self in a service object with mutable state
  - access mutable module-level variables
- Compiler can inform developer where missing locks are preventing a service object or method from being isolated

Part 3
More depth

Providing function arguments by name

```
function foo(int x, int y, int z) {
}

// All these have the same  effect
foo(1, 2, 3);
foo(x = 1, y = 2, z = 3);
foo(z = 3, y = 2, x = 1);
foo(1, z = 3, y = 2);
```

- Arguments can be supplied by name as well as by position
- In a function call, named arguments are transformed into positional arguments using the function's declaration
- Argument names are not part of the function types
- The names of arguments of remote methods and resource methods can be significant
- Changing argument names of public functions is an incompatible change to a module

Error detail

```
error err =
  error("Whoops", httpCode = 27);

type HttpDetail record {
  int httpCode;
};

error<HttpDetail> err =
  error("Whoops", httpCode = 27);

HttpDetail d = err.detail();
```

- An error value contains map containing arbitrary extra details about the error
- Type error<T> describes error value with detail map that has type T
- Named arguments for error constructor specify fields of detail record
- A read-only copy is made of each field using cloneReadOnly
- `error<*>` infers subtype of error from initializer

Error cause

`Example`

- error value has cause of type `error?`
- error(msg, cause) creates error with specified error and cause
- err.cause() gets the cause of an error

Tuples

```
// Fixed length array
type FloatPair float[2];
// Tuple
type FloatPair [float, float];

FloatPair p = [1.0, 2.0];

// Can mix types
type Id [string, int, int];

byte[*] a = base16`DEADBEEF`;
```

- Arrays and tuples are two ways of describing lists
- Tuple is to array as record is to map
- Arrays can be fixed length
- * for length infers fixed length from initializer

Rest type in tuples

```
// int followed by
// zero or more strings
type Id [int, string...];
```

- Already seen `T...` in record types
- `T...` also works in tuples
- `T[]` same as `[T...]`
- Tuples not open by default

Rest parameters

```
function foo(int n, string... s) {
}
```

- s will get bound to array of strings
- Also works in resource path parameters

Spread operator ...x

```
Example
```

- Argument can be map or array
- Works in
  - function call (map or array)
  - list constructor (map)
  - mapping constructor (map)
  - error constructor (map)

Binding patterns

`Example`

- So far _ and record binding patterns
- Can use in assignment
- Specify variable parts of match pattern
- Tuple patterns

Spread in binding patterns

`Example`

- ...x works in binding patterns for
  - lists
  - mappings
  - errors

Default values for record fields

```
type X record {
   string str = "";
};
```

- Record fields can have a default value
- Expression specifying default value must be isolated expression
- Default value does not affect static typing
- cloneWithType (T) will make use of defaults specified by T

Default values for function parameters

```
function search(string str,
                string search,
                int start = 0)
              returns int|() {
   return str.indexOf(search, start);
}
```

- Point 1
- Point 2

`never` type

```
function whoops() returns never {
   panic error("whoops");
}

type InfiniteInts stream<int,never>;

// a record that is open to
// everything except x and y
type NotXY record {
   never x?;
   never y?;
}
```

- No value belongs to the `never` type
- For a function, never means that it cannot return normally
- Other use cases:
  - 2nd type parameter of stream, means stream never ends
  - Open record with optional field of type never is open to everything except that field

readonly type

```
type Coord readonly & record {
   float x;
   float y;
};

// Value of c is immutable
Coord c = {x: 1.0, y: 1.0};

type V readonly & record {
  int[] x;
  int[] y;
};
V v = { x: [1], y: [2] }
// This is OK
readonly & int[] x = v.x;
```

- A value belongs to type `readonly` if it is immutable
- Absence of `readonly` does not guarantee that value is mutable
- For structural type T, use `T & readonly` to describe immutable T
- Structural typing works straightforwardly for readonly structural values: no inherent type is needed

Type intersection

```
type Foo object {
  function foo();
};

type Bar object {
  function bar();
};

type FooBar Foo & Bar;
```

- Type intersection works generally not just for readonly
- T_1 & T_2 means set intersection of types T_1 and T_2
- As well as `readonly`, works for
  - object
  - error

Type intersection for error types

```
type FileErrorDetail record {
  string filename;
};

type FileError
   distinct error<FileErrorDetail>;

type MyError distinct error;

type MyFileError MyError & FileError;
```

- Point 1
- Point 2

Object types

```
type Hashable object {
   function hash() returns int;
};

function h() returns Hashable {
  var obj = object {
     function hash() returns int {
        returns 42;
     }
  };
  // obj belongs to Hashable type
  return obj;
};
```

- Class combines object type with function that can create objects belong to the object type
- Can define object type without implementation
- Object typing is structural
- Analogous to abstract class in Java

Type inclusion for records

```
type Time record {|
   int hour;
   int minute;
   decimal seconds;
|};
type Date record {|
   int year;
   int month;
   int day;
|};
type DateTime record {|
   *Time;
   *Date;
|};
```

- Use *T to include a record type in a record type descriptor
- Effect is similar to copying fields of included record into including record

Included record parameters

```
type Options record {|
   boolean verbose = false;
   string? outputFile = ();
|};

function foo(string inputFile,
             *Options options) {
}

function main() {
   foo("file.text",
        verbose = true);
}
```

- With named arguments
  - function defines each parameter normally
  - caller supplies parameter by name
- With record-typed parameter
  - function uses record for all named parameters
  - caller supplies arguments using mapping constructor
- With included record parameter
  - function defines records for named parameter
  - caller supplies parameter by name
- Named arguments and included record parameters provide consistent experience for caller

Type inclusion for objects and classes

- Constrained so including object is a subtype of every included object

Distinct object types

```
distinct class A {
}

distinct class B {
}

A a = new;

// false
boolean b = a is B;
```

- Objects are structurally typed like everything in Ballerina
- Distinct object types provide similar functionality to nominal typing within a structurally typed framework
- A distinct object value is tagged (branded) with a type-id that is unique to that occurrence of `distinct` in the source
- Both object type and class can be distinct
- Useful for interop with nominally typed object-oriented systems (e.g. Java, GraphQL)

Readonly classes

```
readonly distinct class Date {
   int year;
   int month;
   int day;
   function init(int y, int m, int d) {
      checkDate(y, m, d);
      self.year = y;
      self.month = m;
      self.day = d;
   }
};
```

- class and type
- Point 2

Expression-oriented style

```
function inc(int x) returns int => x + 1;

// same as

function inc(int x) returns int {
   return x + 1;
}

var obj = object {
   private int x = 1;
   function getX() returns int => self.x;
};

// let expressions
function hypot(float x) =>
   let float x2 = x * x in
      float:sqrt(x2 + x2);
```

- Ballerina supports statements for familiarity but also tries to enable an expression-oriented style of programs
- Query expressions and constructors support expression-oriented style
- When function body is an expression can use => instead of block with returns
- Works for methods also
- Let expressions (like let clauses in query expressions) allow you to do more with an expression

Inferring type from context: numeric literals

```
int n = 1; // int
decimal n = 1; // decimal
float n = 1; // float
int|float|decimal n = 1; // int

float n = 1.0; // float
decimal n = 1.0; // decimal
float|decimal n = 1.0; // float
```

- A literal integer can be interpreted as int or float or decimal depending on context; defaults to int
- A literal floating point number can be interpreted as float or decimal depending on context; defaults to float

Interfacing to external code

```
public function open(string path)
               returns handle|error
   = external;
```

- Function body can be defined as `= external`
- `external` keyword can be annotated to say where the implementation comes from
- `handle` type represents opaque handle for use by external functions
  - in a JVM implementation might contain a reference to an object
- `handle` can be wrapped in an object for better type safety
- Alternative is to have an entire module that is implemented in something other than Ballerina

Integer subtypes

`Example`

- int type has built-in subtypes
  - int:Signed32, int:Unsigned32
  - int:Signed16, int:Unsigned16
  - int:Signed8, int:Unsigned8 (same as byte)
- They are subtypes not separate types: work differently from integer types in most other languages
- Useful for interfacing with external systems that use these types

Computed field key

```
const X = "x";
const Y = "y";

map<string> m = {
  [X]: 1,
  [Y]: 2
};
```

- In a mapping constructor, field name can be an expression in square brackets

Additional resources

Implementation: https://ballerina.io/downloads/#swanlake

James Clark's blogs on Ballerina: https://blog.jclark.com/search/label/Ballerina

Online Ballerina docs: https://ballerina.io/swan-lake/learn/

Language specification: https://ballerina.io/spec/lang/draft/latest/

Language issues: https://github.com/ballerina-platform/ballerina-spec/issues
(Use this to ask questions and provide feedback on this presentation)